1974

Life: A Computer Program for Stochastic Life Table Analysis

C. Romesburg

K. Marshall

1973 PROGRESS REPORT

LIFE: A COMPUTER PROGRAM FOR STOCHASTIC LIFE TABLE ANALYSIS

C. Romesburg and K. Marshall
Utah State University

US/IBP DESERT BIOME
RESEARCH MEMORANDUM 74-68

in

REPORTS OF 1973 PROGRESS
Volume 1: Central Office, Modelling
Data Processing Section, pp. 3-8

MAY, 1974

Citation format: Author(s). 1974. Title.
US/IBP Desert Biome Res. Memo. 74-68.
Utah State Univ., Logan. 6 pp.

Ecology Center, Utah State University, Logan, Utah 84322

INTRODUCTION

The life table analysis computer program LIFE was coded by the Desert Biome Data Processing Group in response to a principal investigator's request for data analysis. The program handles both current and cohort life tables in abridged and unabridged form, and generates standard errors for three basic life table parameters. We found programs available in the open literature -- such as Biomedical Computer Programs (BMD) life table analysis -- to give a less general treatment to life table data. The purpose of this report is to make the program more widely available to the scientific community.

PROGRAM DESCRIPTION

Life tables may be either current or cohort type. A current life table requires the age structure of organisms and the number of deaths in each age-class as basic input data. These data are a summary of the mortality experience of a given population at one short instance in time. Census data for the United States population where people are classified according to age and whether or not they die in the given year of census constitute current life-table data. On the other hand, a cohort life table follows one population from birth until no member of the population is living, recording the years in which the individual members die. An example would be the life dynamics of a laboratory-reared insect hatch.

Either current or cohort life tables may be unabridged or abridged. If unabridged, the basic unit of time for classification of death is taken in equal increments throughout the life cycle: 0-1, 1-2, 2-3, etc. The units are arbitrary and depend to a large extent on the life span of the organism; years may be used for people while days may be more appropriate for insects. An abridged life table allows age intervals greater than the basic unit and, in general, these may be unequal. For example, 0-1, 1-5, 5-10, 10-20, etc.

It should be noted that the entities need not be organisms. For example, if a cohort of books in a library is followed through time until they are destroyed, lost or stolen, a life table of book mortality may be meaningful. Also, in some instances time may not be the independent variable. By example, suppose a test part made from a metal alloy is subjected to a bending fatigue test until the part breaks. The life table may then appear with "age-classes" given by the number of cycles prior to failure: 0-6000, 6000-12,000, 12,000-18,000, etc.

The life tables calculated by LIFE are according to a binomial probability model detailed in Chiang (1968, Ch. 9-10). Two previous papers by Chiang (1960a, 1960b) contain additional relevant material. This report gives a correspondence between the notation used by Chiang and FORTRAN variables used in LIFE. Also included are the results of two test problems taken from Chiang (1968, 1960b); the reader will gain substantial understanding by comparing these outputs with the originals.

TERMINOLOGY

The following is the correspondence between symbols used by Chiang and the computer program LIFE. Type of variable is either I = integer or F = decimal.

Symbol: Chiang	Symbol: LIFE	Variable Type	Definition
—	ITYPE	I	Type of life table (current or cohort)
w	NUM	I	Number of age-classes
x_i	X	I	Lower age of age interval X(I) — X(I+1)
P_i	P	I	Mid-interval population in age interval
D_i	D	I	Number of deaths in age interval referenced to population P
l_0	RADIX	I	RADIX; 100000 is default value
l_i	L	I	Number living at age X(I)
d_i	DD	I	Number of deaths referenced to L
a_i	A	F	Fraction of last age interval of life for age X(I)
M_i	M	F	Death rate in age interval

Table, continued

Symbol		Type of Variable	Definition
Chiang	LIFE		
q_i	Q	F	Proportion dying in age interval
Sq_i	SQ	F	Sample standard error of Q
P_{oi}	POI	F	Proportion of survivors over age interval X(0) — X(I)
Sp_{oi}	SPOI	F	Sample standard error of POI
e_i	E	F	Observed expected life at age X(I)
Se_i	SE	F	Sample standard error of E

INPUT

For a current life table, values for X, P, D, A, and RADIX must be read in. To illustrate, if there are 10 age-classes (NUM = 10), the program will expect 10 values for each of X, P and D. The variable A -- the fraction of last age interval of life -- is programmed to default to 0.5 if no values of A are read in. The reason for this is that in most cases it is reasonable to expect that when many organisms die in a given age interval, the fraction of the interval lived when death occurs is one-half, or 0.5. An instance when this is sometimes not reasonable is when considering death among very young organisms. For example, death among human infants in the first year of life often occurs shortly after birth. When all infants dying in the first year of life are considered we find that, on the average, approximately, say, one-tenth of the first year is lived. Be this the case, a value of 0.1 for A is more appropriate for the first year of life.

The user can override the default value by reading in A values as follows. The order in which the A's are read in is, correspondingly, from youngest to oldest age intervals. For example, if three values of A are read in these will be assigned in order to the first, second and third intervals of the life table; all values for intervals greater than the third will default to 0.5.

A final clarification concerns the variable RADIX. A current life table contains a column labeled l_i (or L) which is the "number living at age X(I)." The numbers in this column are based on an artificial population in which the number of organisms in the first age interval is arbitrarily set equal to the variable RADIX. The death rates obtained from the input life-table data are then used to obtain the l_i values for the remaining, older age intervals. Thus, the l_i column represents the mortality pattern obtained from the input data as imposed on a fictitious population. The LIFE program uses a default value of RADIX = 100000. This default value can be overridden to the extent that the user is given the option of specifying a smaller value (less than 100000).

Input format for LIFE follows. Integer entries are symbolized by I. Right-adjust integers within field. Decimal variables are denoted by F. If the decimal is punched they can be located anywhere within the field; otherwise, observe the given format.

Whether a given life table is current or cohort type is specified on Card Type I. The choice of abridged or unabridged life table is implicit in the statement of age-class intervals on Card Type III. There is one card per case for Card Type I. Card Type II is used only if the user is reading in A values. If a given card of Card Type II becomes filled, continue onto a subsequent card. There is one Card Type III card for each age-class interval. These are ordered from youngest to oldest ages.

Card Type I

Columns	Variable Type	Description
5	I	ITYPE. Enter type of life table: 0 — current; 1 — cohort
6 — 10	I	NUM. Enter number of age intervals
11 — 15	I	IA. Number of A values to be read in
16 — 20	I	RADIX. An entry of 0 causes the default value RADIX = 100000 to be used. If a smaller value is desired it should be entered. RADIX is only required for a a current life table. If life table is cohort type, there will be no entry for RADIX.

Card Type II (use this card only if IA on Card Type I is greater than zero. Format is 20F 4.3)

Columns	Variable Type	Description
1 — 4	F	The value of A for the first age interval
5 — 8		The value of A for the second age interval
9 — 12		The value of A for the third age interval
.		.
.		.
.		.
76 — 80		.

Card Type III (one card for each interval)

Columns	Variable Type	Description
1 — 5	I	X(I) Lower age of age interval, X(I) — X(I+1). Note that when I=1, X(1)=0
11 — 20	I	If life table is current (ITYPE=0), enter P, "mid-interval population in age interval." If life table is cohort (ITYPE=1), enter L, "number living at age X(I)."
21 — 30	I	If life table is current (ITYPE=0), enter D, "number of deaths in age interval referenced to population P." If life table is cohort (ITYPE=1), no entry is required.

TEST PROBLEMS

Two test problems are shown below. The first is an abridged current life table for the total 1960 United States population. This example is worked out in Chiang (1968, Ch. 9, Tables 4, 5, 6, 7, and 8). The second example from Chiang (1960b) is an abridged cohort life table for male *Drosophila melanogaster* raised in the laboratory. For each test problem, a listing of the input cards and a listing of the output are shown.

INPUT LISTING FOR ABRIDGED CURRENT LIFE TABLE

Card Type I:

```
0  21  20
```

Card Type II:

```
010 039 046 054 057 049 050 052 054 054 054 053 052 052 052 051 051 048 045 041
```

Card Type III:

```
 0   4126560   110873
 1  16195304    17682
 5  18659141     9163
10  16815965     7374
15  13287434    12185
20  10803165    13348
25  10870386    14214
30  11951709    19200
35  12508316    29161
40  11567216    42942
45  10928878    64283
50   9696502    90593
55   8595947   116753
60   7111897   153444
65   6186763   196605
70   4661136   223707
75   2977347   219978
80   1518206   185231
85    648581   120366
90    170653    50278
95     44551    13882
```

OUTPUT FOR ABRIDGED CURRENT LIFE TABLE

I	X(I)-X(I+1)		P	D	M	A	Q	SQ	L	DD	E	SE	PDI	SPDI
1	0	1	4126560	110873	0.02687	0.10	0.02623	0.000078	100000	2623	69.65	0.01177	1.000	0.00000
2	1	5	16195304	17682	0.00109	0.39	0.00436	0.000033	97377	424	70.53	0.01065	0.974	0.00008
3	5	10	18659141	9163	0.00049	0.46	0.00245	0.000026	96952	238	66.83	0.01045	0.970	0.00008
4	10	15	16815965	7374	0.00044	0.54	0.00219	0.000025	96715	212	61.99	0.01035	0.967	0.00009
5	15	20	13287434	12185	0.00092	0.57	0.00458	0.000041	96503	442	57.12	0.01026	0.965	0.00009
6	20	25	10803165	13348	0.00124	0.49	0.00616	0.000053	96061	592	52.37	0.01005	0.961	0.00010
7	25	30	10870386	14214	0.00131	0.50	0.00652	0.000054	95470	622	47.68	0.00975	0.955	0.00011
8	30	35	11951709	19200	0.00161	0.52	0.00800	0.000058	94848	759	42.97	0.00950	0.948	0.00012
9	35	40	12508316	29161	0.00233	0.54	0.01159	0.000068	94089	1091	38.30	0.00928	0.941	0.00013
10	40	45	11567216	42942	0.00371	0.54	0.01840	0.000088	92998	1712	33.72	0.00906	0.930	0.00015
11	45	50	10928878	64283	0.00588	0.54	0.02902	0.000113	91286	2649	29.30	0.00878	0.913	0.00016
12	50	55	9696502	90593	0.00934	0.53	0.04571	0.000148	88637	4052	25.09	0.00847	0.886	0.00019
13	55	60	8595947	116753	0.01358	0.52	0.06577	0.000186	84586	5563	21.17	0.00808	0.846	0.00022
14	60	65	7111897	153444	0.02158	0.52	0.10257	0.000248	79023	8105	17.48	0.00769	0.790	0.00026
15	65	70	6186763	196605	0.03178	0.52	0.14763	0.000307	70917	10470	14.18	0.00725	0.709	0.00031
16	70	75	4661136	223707	0.04799	0.51	0.21472	0.000402	60448	12979	11.18	0.00695	0.604	0.00034
17	75	80	2977347	219978	0.07388	0.51	0.31280	0.000553	47468	14848	8.54	0.00683	0.475	0.00036
18	80	85	1518206	185231	0.12201	0.48	0.46312	0.000788	32620	15107	6.27	0.00704	0.326	0.00036
19	85	90	648581	120366	0.18558	0.45	0.61437	0.001100	17513	10760	4.60	0.00774	0.175	0.00032
20	90	95	170653	50278	0.29462	0.41	0.78812	0.001618	6754	5323	3.36	0.00996	0.068	0.00023
21	95		44551	13882	0.31160		1.00000		1431	1431	3.21		0.014	0.00012

INPUT LISTING FOR ABRIDGED COHORT LIFE TABLE

Card Type I:

1	13	0	0

Card Type II:

0	270	2
5	268	4
10	264	3
15	261	7
20	254	3
25	251	3
30	248	16
35	232	66
40	166	36
45	130	54
50	76	42
55	34	21
60	13	13

OUTPUT FOR ABRIDGED COHORT LIFE TABLE

I	X(I)-X(I+1)		L	DD	A	Q	SQ	E	SE	POI	SPOI
1	0	5	270	2	0.50	0.00741	0.005218	43.19	0.69929	1.000	0.00000
2	5	10	268	4	0.50	0.01493	0.007407	38.49	0.67074	0.993	0.00522
3	10	15	264	3	0.50	0.01136	0.006523	34.03	0.62303	0.978	0.00897
4	15	20	261	7	0.50	0.02682	0.010000	29.40	0.59401	0.967	0.01092
5	20	25	254	3	0.50	0.01181	0.006779	25.14	0.54028	0.941	0.01437
6	25	30	251	3	0.50	0.01195	0.006859	20.41	0.52367	0.930	0.01557
7	30	35	248	16	0.50	0.06452	0.015600	15.63	0.51485	0.919	0.01665
8	35	40	232	66	0.50	0.28448	0.029621	11.53	0.49815	0.859	0.02116
9	40	45	166	36	0.50	0.21687	0.031986	10.12	0.46017	0.615	0.02962
10	45	50	130	54	0.50	0.41538	0.043220	7.23	0.43280	0.481	0.03041
11	50	55	76	42	0.50	0.55263	0.057035	5.59	0.43607	0.281	0.02737
12	55	60	34	21	0.50	0.61765	0.083342	4.41	0.41671	0.126	0.02019
13	60		13	13		1.00000		2.50		0.048	0.01303

SUMMARY

The main value of Chiang's (1968) life table model is the standard errors generated for q_i, p_{oi} and e_i. These are useful for assessing the inherent uncertainty in the estimators. The standard errors also make possible tests of hypotheses concerning life tables for two or more populations. For example, the hypothesis that the true expected life at a given age is the same for two populations can be tested. Chiang (1960b) gives a worked example of the hypothesis testing procedure.

LITERATURE CITED

CHIANG, C. L. 1960a. A stochastic study of the life table and its application: I. Probability distributions of the biometric function. Biometrics 16:618-635.

CHIANG, C. L. 1960b. A stochastic study of the life table and its applications: II. Sample variance of the observed expectation of life and other biometric functions. Human Biol. 32:221-238.

CHIANG, C. L. 1968. Introduction to stochastic processes in biology. John Wiley and Sons, New York.

APPENDIX 1

FORTRAN Listing of LIFE Program

```
FILE  5=FILE5,UNIT=DISKPACK,RECORD=14,BLOCKING=30
FILE  6=FILE6,UNIT=PRINTER,RECORD=22
      REAL A(100),M(100),L(100),DD(100),Q(100),SQ(100),
     1 POI(100),SPOI(100),E(100),SE(100)
      INTEGER X(100),P(100),D(100)
 1    READ(5,100,END=5) ITYPE,NUM,IA,RADIX
      IF (RADIX .EQ. 0) RADIX=100000
      DO 2 I=(IA + 1),NUM
 2    A(I)=0.5
      IF (IA .GT. 0) READ(5,101) (A(I),I=1,IA)
      READ(5,102) (X(I),P(I),D(I),I=1,NUM)
      CALL LIFE(ITYPE,NUM,RADIX,X,A,P,D,L,DD,M,Q,SQ,POI,SPOI,E,SE)
      IF (ITYPE .EQ. 1) GO TO 4
      WRITE(6,300)
      WRITE(6,200) (I,X(I),X(I+1),P(I),D(I),M(I),A(I),Q(I),SQ(I),L(I),
     1 DD(I),E(I),SE(I),POI(I),SPOI(I),I=1,(NUM - 1))
      WRITE(6,201) NUM,X(NUM),P(NUM),D(NUM),M(NUM),Q(NUM),L(NUM),DD(NUM)
     1 ,E(NUM),POI(NUM),SPOI(NUM)
      GO TO 1
 4    WRITE(6,301)
      WRITE(6,202) (I,X(I),X(I+1),L(I),DD(I),A(I),Q(I),SQ(I),
     1 E(I),SE(I),POI(I),SPOI(I),I=1,(NUM - 1))
      WRITE(6,203) NUM,X(NUM),L(NUM),DD(NUM),Q(NUM),E(NUM),
     1 POI(NUM),SPOI(NUM)
      GO TO 1
 5    STOP
100   FORMAT(4I5)
101   FORMAT(20F4.3)
102   FORMAT(I5,2I10)
200   FORMAT(3I5,2I10,F10.5,F5.2,F10.5,F10.6,2I8,F8.2,F10.5,F8.3,F10.5)
201   FORMAT(2I5,5X,2I10,F10.5,F15.5,10X,2I8,F8.2,10X,F8.3,F10.5)
202   FORMAT(3I5,2I10,F6.2,F10.5,F10.6,F8.2,F10.5,F8.3,F10.5)
203   FORMAT(2I5,5X,2I10,F16.5,10X,F8.2,10X,F8.3,F10.5)
300   FORMAT('1    I X(I)-X(I+1)        P          D         M    A       Q
     1      SQ          L         DD         E          SE        POI      SPOI '
     2 / ' -' 20('-------'))
301   FORMAT('1    I X(I)-X(I+1)        L          DD       A         Q
     1SQ        E          SE         POI      SPOI ' / ' ',16('-------'))
      END
```

```
      SUBROUTINE LIFE(ITYPE,NUM,RADIX,X,A,P,D,L,DD,
     1 M,Q,SQ,POI,SPOI,E,SE)
      INTEGER X(100),P(100),D(100)
      REAL A(100),L(100),DD(100),M(100),Q(100),
     1 SQ(100),POI(100),SPOI(100),E(100),SE(100)
C
C INTEGER VARIABLE DEFINITIONS
C ITYPE - TYPE OF TABLE: 0=CURRENT, 1=COHORT
C NUM   - NUMBER OF AGE CLASSES
C X     - LOWER AGE OF AGE INTERVAL X(I) - X(I+1)
C P     - MID-INTERVAL POPULATION IN AGE INTERVAL
C D     - NUMBER OF DEATHS IN AGE INTERVAL
C            REFERENCED TO POPULATION P
C RADIX - RADIX; 100000 DEFAULT
C L     - NUMBER LIVING AT AGE X
C DD    - NUMBER OF DEATHS REFERENCED TO L
C
C REAL VARIABLE DEFINITIONS
C A     - FRACTION OF LAST AGE INTERVAL OF LIFE
C            FOR AGE X
C M     - DEATH RATE IN AGE INTERVAL
C Q     - PROPORTION DYING IN AGE INTERVAL
C SQ    - SAMPLE STANDARD ERROR OF Q
C POI   - PROPORTION OF SURVIVORS OVER AGE INTERVAL
C SPOI  - SAMPLE STANDARD ERROR OF POI
C E     - OBSERVED EXPECTION OF LIFE AT AGE X
C SE    - SAMPLE STANDARD ERROR OF E
C
C CURRENT LIFE TABLE
C   INPUT:  NON-ARRAY VARIABLES: ITYPE,NUM,RADIX
C           ARRAY VARIABLES: X, P, D, A
C
C   OUTPUT: ARRAY VARIABLES: M, Q, SQ, L, DD, POI,
C                            SPOI, E, SE
C
C
C COHORT LIFE TABLE
C   INPUT:  NON-ARRAY VARIABLES: ITYPE, NUM
C           ARRAY VARIABLES: X, L, A
C
C   OUTPUT: ARRAY VARIABLES: Q, SQ, L, D, POI, SPOI,
C                            E, SE
C
      IF (ITYPE .EQ. 1) RADIX=P(1)
      IF (RADIX .EQ. 0) RETURN
      DO 3 I=2,NUM
        J=I - 1
        XN=X(I) - X(J)
        IF (ITYPE .EQ. 1) GO TO 1
        M(J)=FLOAT(D(J))/FLOAT(P(J))
C
C     COMPUTE POPULATION AT START OF INTERVAL
C
        POP=FLOAT(P(J))/XN+FLOAT(D(J))*(1.0 - A(J))
        GO TO 2
 1      D(J)=P(J) - P(I)
        POP=P(J)
 2      Q(J)=FLOAT(D(J))/POP
        SQ(J)=Q(J) * (1.0 - Q(J))/POP
        L(J)=RADIX
        DD(J)=RADIX * Q(J)
        RADIX=RADIX - DD(J)
 3    CONTINUE
      Q(NUM)=1.0
      M(NUM)=FLOAT(D(NUM))/FLOAT(P(NUM))
      L(NUM)=RADIX
      DD(NUM)=RADIX
      IF (ITYPE .EQ. 1) GO TO 4
      TYEARS=L(NUM)/M(NUM)
      GO TO 5
 4    TYEARS=FLOAT(P(NUM)) * A(NUM) *
     1 (X(NUM) - X(NUM - 1))
 5    E(NUM)=TYEARS/L(NUM)
      SVAR=0
      DO 6 I=1,(NUM - 1)
        J=NUM - I
        XN=X(J + 1) - X(J)
        TYEARS=TYEARS + XN * (L(J) + DD(J) *
     1    (A(J) - 1.0))
        E(J)=TYEARS/L(J)
        SVAR=SVAR + L(J) * L(J) * SQ(J) *
     1    (E(J + 1) + XN * (1.0 - A(J)))**2
        SE(J)=SVAR/(L(J) * L(J))
 6    CONTINUE
      SVAR=0
      XN=1.0/L(1)
      DO 7 I=1,(NUM - 1)
        POI(I)=L(I) * XN
        SPOI(I)=SQRT(POI(I) * POI(I) * SVAR)
        SVAR=SVAR + SQ(I)/(1.0 - Q(I))**2
        SQ(I)=SQRT(SQ(I))
        SE(I)=SQRT(SE(I))
 7    CONTINUE
      POI(NUM)=L(NUM)* XN
      SPOI(NUM)=SQRT(POI(NUM) * POI(NUM) * SVAR)
      RETURN
      END
```